
BY LEO J .  LANGEVIN

INPart I (Technical Support, January 2000) I presented a
simple VSE system application called $DA. Not only

did this application provide a replacement for a facility that
IBM discarded with the release of VSE/ESA, but $DA serves
as an example of the number of ways that you can convert a
local application to a web-based solution.

As I indicated in Part I, you can either convert the program
to a Common Gateway Interface (CGI) or convert the program
into a server. With the latter method there are two variations:

1. You can have a server program pretend it’s a user and 
invoke the old program, get the result, format the data,
and return it to the web browser. This will require few
program changes.

2. Or, you can have the server perform the function itself 
and return the data to the web browser. This may require
moderate program changes.

This series is not about using screen-scraper programs or
providing 3270-access through the web; rather it is about
eliminating the use of a 3270 client altogether. Migrating an
application to the web is not labor-intensive and when compared
to screen-scraping options, it is a superior methodology.
Also, when compared to 3270-to-HTML-to-3270 translation
interfaces, it’s much faster!

So where were we?
The $DA application presented in Part I is fairly simple.

You invoke it and it displays a screen. You press Enter and the
screen refreshes. You type any non-blank character and press
Enter and the application terminates. There are columns of
data that are static and there are rows of data that are dynam-
ic (the number of lines on the screen grows and shrinks,
depending on system activity).

We could design our Internet application to operate in the
same way and make no other changes, but then, what’s the
point of migrating $DA to the web? Instead of maintaining
the status quo, we’ll make a few minor enhancements. For
example, the old application is without color, has fixed length
fonts, and while it does provide a great function, it clearly
lacks a nice looking format. So in our “webified” version of
the same application, let’s add a touch of color and provide
nice looking borders around each column. 

You may want to enhance this even further. For example,
you may want dynamic partitions to use a different color. Or,
perhaps cause a job name to blink if it’s not VTAM, CICS,
TCP/IP or any other long running job, but has been running
for more than an hour. You may want to cause the following
“meta” command to be passed back to the browser:

<META HTTP-EQUIV=”Refresh” CONTENT=”10”; URL=$da”>

This will tell the web browser to re-issue the “$DA” command
every 10 seconds, causing the screen to be refreshed auto-
matically.

These are simple cosmetic changes. Usually, someone will
make very few changes to the original application and then
add these cosmetic changes once they get the application
fully functional.

Before we begin creating our CGI, let’s convert
“CALL$DA” to “CALL$DAH”. Why? Well, if you remember
from Part I that “CALL$DA” would simply send back lines
of information to the main part of the code, the new version,
CALL$DAH, will not just send back lines of information, but
will also include pre-formatted HTML commands to create
the cosmetic look that we want. As you can see from Figure
1, the logic of the code remains the same; only the layout of
the returned record has been modified. 

Migrating VSE System
Applications to the Web: Part II

— Porting the Program 
to Act as a CGI   

This article examines how to port the simple program $DA, which was created in the first article,
to act as a Common Gateway Interface for web access.

* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *   
* Program: CALL$DAH                                 V 2.0  *   
*                                                     *   
* Purpose: To display partition information like “$DA” with    *   
*          HTML parameters for formatting.                    *   
*                                                         *   
* INPUT: Parameter in R1 points to buffer address pointer and       *   
*        a fullword receiving field which will contain the          *   
*        number of lines returned by this routine.                 *   
*                                                          *   
* Subroutines:  DSPLHEX, CNVTSTCK, TIMEDIFF                         *   
* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *   
CALL$DAH SUBRT VRM=2.0,AMODE=31 Begin subroutine               

AMODESW SET,AMODE=31,WR=(11),SAVE=(7) 
ST    R7,HOLDAMOD      Save the value                
LM    R3,R4,0(R1)        Get the input/output pointers   
LR    R11,R3          Point to the beginning         
MODESET KEY=ZERO         Let’s not blow up! 
USING COMREG,R5        Map to the COMREG layout       
COMRG ,              Get the current COMREG pointer  
LR    R5,R1           We’ll save that                
MVC   TOD(9),MASK       Prepare for the display        
GETIME STANDARD        Get the current time of day    
ST    R1,WORK           Save this value                
ED    TOD(9),WORK       Unpack/Format the time         
MVI   TOD,C’ ‘         Remove the initial byte        
MVC   0(256,R3),HTMLHEAD  Get the first part             
MVC   256(256,R3),HTMLHEAD+256 Get the next part              
A     R3,HEADLEN        Skip to the detail area        
XR    R7,R7             Point to low core              
USING SYSS00,R7        Map to lowcore layout          
L     R7,APCBATAB        Get the PCB table pointer      
USING PCBADR,R6         Map to PCB/PCE layout          
USING PIBADR,R2          Map to PIB layout              
STCK  CURRTIME       Get the current time           
CALL  CNVTSTCK,(CURRTIME)  Convert for later use          

LOOP     LA   R7,4(R7)           Point to next PCBATAB entry    
CLC  0(4,R7),ENDTABLE   END OF TABLE ? 
BZ   LM0300            YES...RETURN TO CALLER         
CLC  0(4,R7),INACTIVE   INACTIVE DYNAMIC ENTRY ? 
BZ   LOOP             YES...SKIP IT                 
L    R6,0(R7)          GET THE PCB                   
L    R5,PCECOMRA         GET THE COMREG                
CLC  COMNAME(8),=CL8’NO NAME’ INACTIVE PARTITION ? 
BZ   LOOP               TRUE...SKIP IT                 

* PARTITION ID                                                          
MVC  ID(2),IJBSPNLI      PUT IN THE PARTITION ID        

* GET PIB VALUE                                                        
GETFLD FIELD=PIB,LOGID=ID  GET THE PIB                   
LR   R2,R0              SAVE THE PIB                  
XC   WORK(4),WORK       INIT THE VALUE                
MVC  WORK+3(1),PIBFLG    GET THE PIB                   
CALL DSPLHEX,(WORK)       MAKE IT DISPLAYABLE            
MVC  PIB(2),WORK+10     GET THE RETURNED VALUE         
XC   PIDHOLD(2),PIDHOLD  INIT THE PIK TO BG             
CLC  ID(2),=C’BG’        IS IT IN BG PARTITION ? 
BZ   USEPID              YES...TAKE THE DEFAULT         
MVC  PIDHOLD(2),PID     NO...PUT IN THE PIK FROM COMREG 

USEPID   DS    0H                                                   
GETJA PART=PIDHOLD,ACTION=UPDATE                             
L    R9,JAPART       POINT TO JOB ACCOUNTING TABLE   

* DOS JOB NAME                                                         
MVC  DNAME(8),COMNAME   GET THE DOS NAME               

* PHASE NAME                                                           
MVC  PNAME(8),IJBPHNAM GET THE PHASE NAME             

* GET THE NUMBER OF SIO’S                                               
SR    R14,R14             INIT THE WORK COUNTER          
LH    R10,12(R9)         GET THE NUMBER OF SIO’S        
LA    R8,120(R9)         POINT TO SIO FIELDS            
LTR   R10,R10           IS THERE AT LEAST 1 SIO ? 
BZ    LM0050           NO...SKIP IT                  

SIOLOOP  TM    0(R8),X’20’      GONE TOO FAR ? 
BO    SHOWSIO         YES...TERMINATE                
A     R14,2(R8)       UPDATE TOTAL COUNT             
LA    R8,6(R8)          POINT TO NEXT SIO GROUP        
BCT   R10,SIOLOOP       AND GET THOSE VALUES           

SHOWSIO  CVD   R14,DOUBLE         CONVERT SIO TOTAL TO DECIMAL    
MVC   SIO(10),DMASK     GET THE EDIT MASK              
ED    SIO(10),DOUBLE+4    MAKE IT DISPLAYABLE            

* GET THE ELAPSED TIME                                                  
MVC   OLDTIME(8),16(R9)   GET THE START TIME             
CALL  CNVTSTCK,(OLDTIME)  CONVERT FOR CALCULATIONS       
CALL  TIMEDIFF,(TDPARM)   GET THE DIFFERENCE             
MVC   ELAPSED(9),TDRESULT  AND SAVE IT                   

FIGURE 1: CALL$DAH

* CPU TIME                                                             
SR   R0,R0          INIT HIGH VALUE REG            
L    R1,28(R9)          GET THE CPU TIME              
D    R0,=F’3’       CONVERT TO DISPLAY             
CVD  R1,DOUBLE         MODIFY FOR THE DISPLAY         
MVC  CPU(12),CMASK      SETUP THE EDIT STRING          
ED   CPU(12),DOUBLE+3    MAKE IT DISPLAYABLE            

* RETURN A LINE OF INFO                                                 
LM0050   MVC   0(DETAILEN,R3),DETAIL Get the detail lines           

LA    R3,DETAILEN(,R3)  Skip to the next output        
B     LOOP             DO IT FOR EVERYONE            

LM0300   DS    0H                                                   
MODESET KEY=USER        MODIFY PROTECTION KEY TO NON-0 
MVC   0(FOOTLEN,R3),FOOTER GET THE HTML FOOTER COMMAND    
LA    R3,FOOTLEN(,R3)   Skip ahead                    
SR    R3,R11             Get the size                  
ST    R3,0(R4)           Pass back the value            
L     R7,HOLDAMOD       Regain the AMODE              
AMODESW SET,AMODE=(7)    RESET THE AMODE               
SUBRT ,               RETURN TO CALLER              

*  — — — — — — — — — — — — — — DATA AREA — — — — — — — — — — — — —  *             
PIDHOLD  DC    H’00’                                                
HOLDAMOD DS    F                                                   
*                                                                      

DS    0D                                                   
TDPARM   DS    0CL25       TIME DIFFERENCE PARAMETER      
OLDTIME  DS    D              EARLIER TIME FOR LAPSE CALC    
CURRTIME DS    D                CURRENT TIME FOR LAPSE CALC    
TDRESULT DS    CL9             DIFFERENCE BETWEEN OLD AND NEW 
*                                                                      
HTMLHEAD DC  C’<HTML><BODY BGCOLOR=BLACK TEXT=ORANGE>’                 

DC  C’<TABLE BORDER=5><TR><TH>’                              
TOD      DC  CL13’ ‘                                                

DC  C’</TH></TR></TABLE>’                                    
DC  C’<TABLE BORDER=2><TR>’                                  
DC  C’<TH><PRE>ID</TH>’                                      
DC  C’<TH><PRE>PIB</TH>’                                     
DC  C’<TH><PRE>Job Name</TH>’                                
DC  C’<TH><PRE>Phase Name</TH>’                              
DC  C’<TH><PRE>CPU</TH>’                                     
DC  C’<TH><PRE>Elapsed</TH>’                                 
DC  C’<TH><PRE>SIO</TH>’                                     
DC  C’</TR><TR>’                                            

HEADEND  DS  0F                                                     
HEADLEN  DC  AL4(HEADEND-HTMLHEAD) 
*                                                                      
DETAIL   DS   0H                                                     
IOAREA   DC   CL10’<TD><PRE>’                                         
ID       DC   CL02’ ‘                                                

DC   CL09’<TD><PRE>’                                         
PIB      DC   CL02’ ‘                                                

DC   CL09’<TD><PRE>’                                         
DNAME    DC   CL08’ ‘                                                

DC   CL09’<TD><PRE>’                                         
PNAME    DC   CL08’ ‘                                                

DC   CL21’<TD ALIGN=RIGHT><PRE>’                              
CPU      DC   CL12’ ‘                                                

DC   CL21’<TD ALIGN=RIGHT><PRE>’                              
ELAPSED  DC   CL09’ ‘                                                

DC   CL21’<TD ALIGN=RIGHT><PRE>’                              
SIO      DC   CL10’ ‘                                                

DC   CL9’</TR><TR>’                                          
DETAILEN EQU  *-DETAIL                                                
*                                                                      
FOOTER   DC  C’</TR></TABLE></PRE></BODY></HTML>’                       
FOOTLEN  EQU  *-FOOTER                                                
*                                                                      
* WORK FIELDS                                                           
ENDTABLE DC   X’FFFFFFFF’      END OF TABLE INDICATOR        
INACTIVE DC   X’00000000’       INACTIVE DYNAMIC PARTITION    
DOUBLE   DS   D                WORK FIELD FOR CVD           
WORK     DS   3F              PARM FOR DSPLHEX             
MASK     DC   X’2120207A20207A2020’ TOD WORK FIELD               
CMASK    DC   X’40202020206B2020214B2020’ FOR TIMERS                  
DMASK    DC   X’40206B2020206B202021’ MASK FOR DECIMAL SIZES        
*  — — — — — — — — — — — — DUMMY SECTIONS — — — — — — — — — — — — — —  *           

DROP R12,R5,R6,R2,R7                                          
MAPPCB ,                PCB/PCE LAYOUT               
MAPPIB  PIB           PIBTAB LAYOUT                
MAPCOMR ,               COMREG LAYOUT                
SGLOWC DSECT=YES        LOWCORE LAYOUT               
END    

CONTINUED

PRINT NOGEN                                                   
PUNCH ‘ PHASE CGI$DA,*’                                        

* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *     
* Program: CGI$DA                                    *     
*                                                          *     
* Purpose: Demonstrate how to code a sample CGI: TCP/IP for VSE.  *     
*          This CGI will:                                 *     
*                                                        *     
*          1) Send back a formatted “$DA” display to the user.  *     
*                                                        *     
* Description: This program is for demonstration purposes on how  *     
*          to code a CGI.                                    *     
* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *     
SAVE DSECT                                                         

DS 3F                                                     
SVER14   DS    F                                                      
SVER15   DS    F                                                      
SVER00   DS    F                                                      
SVER01   DS    F                                                      
SVER02   DS    F                                                      
SVER03   DS    F                                                      
SVER04   DS    F                                                      
SVER05   DS    F                                                      
SVER06   DS    F                                                      
SVER07   DS    F                                                      
SVER08   DS    F                                                      
SVER09   DS    F                                                      
SVER10   DS    F                                                      
SVER11   DS    F                                                      
SVER12   DS    F                                                      
SVER13   DS    F                                                      

FRBLOK DSECT=YES
*                                                                      
* DEFINE THE CGI HEADER (LIST OF ROUTINES) - REQUIRED!!! 
*                                                                      
CGI$DA FILEIOHD DEFINE,     DEFINE A FILE HANDLER      X

BEGIN=BEGIN,     HANDLER STARTUP      X
FINISH=FINISH,   HANDLER SHUTDOWN        X
OPEN=OPEN,         FILE OPEN          X
CLOSE=CLOSE,      FILE CLOSE          X
WRITE=0,        FILE WRITE         X
READ=READ          FILE READ                        

*                                                                      
* SECTION: READ                                                         
* PURPOSE: Return a web page to the client. 
*                                                                      
* INCOMING REGS: 
*  R0  = Length of the buffer                                         
*  R1  = Buffer address for returning the data                        
*  R2  = ACBLOK pointer                                              
*                                                                      
* The ACBLOK has a layout as follows: 
* INDICATORS  DC F’0’                                            
* BUFFPOINTER DC A(BUFFER) 
* SIZEPOINTER DC A(SIZEHOLDARE) 
* SIZEHOLDARE DC F’0’                                            
* BUFFER    DS CL4080                                          
*                                                                      
* OUTGOING REGS: 
*  R0   - Buffer length                                           
* R15=0 - Ok                                                     

R15=4 - All done                                               
*                                                                      
READ  SUBRT TYPE=BEGIN,PHASE=NO,CATALR=NO                            

USING SAVE,R8        Map to the DSECT               
L    R8,4(R13)           Point to old save area         
LR   R6,R1                We’ll redirect it              
XR   R9,R9               Clear the counter              
LR   R5,R0              Maximum length                 
CLI  0(R2),X’FF’        Already do a mime ? 
BZ   READDO             Yes...proceed                  
CLI  0(R2),X’FE’        Already do the body ? 
BZ   READEOF            Yes...exit                     

* Send the mime header                                                  
MVC 0(9,R6),=C’text/html’ Insert the mime header         
MVC  9(2,R6),=X’0D25’    With a CR/LF                   
L    R9,=F’11’            Here is the length             
MVI  0(R2),X’FF’          Setup for next time            
XR   R15,R15            Clear the return code          
B    READOUT             So exit                       

* Send the body of data                                                 
READDO   LA   R1,4(R2)             Point to the parameters        

L    R15,=V(CALL$DAH)     Point to the routine           
BALR R14,R15            Perform the routine            
L    R9,12(R2)           Get the number of bytes        
L    R14,4(R2)           Point to the data              
LR   R4,R9               For looping                    

$DALOOP  LTR  R4,R4               All done ? 
BNP FILLED               Yes...exit                     

FIGURE 2: CGI$DA

MVC  0(256,R6),0(R14)    Return the data                
LA   R6,256(,R6)          Skip the output                
LA   R14,256(,R14)         Skip the input                 
S    R4,=F’256’          Deduct the rest                
B    $DALOOP              And repeat                    

FILLED   XR   R15,R15             Indicate ok                    
MVI  0(R2),X’FE’         For the last time through       
B    READOUT            And exit                      

*                                                                      
READEOF  L    R15,=F’4’           Indicate all done              

XR   R9,R9              Return nothing                 
MVI  0(R2),0             Clear the indicator            

*             
READOUT  ST  R9,SVER01           Save this value                

SUBRT RC=0(R15)         Return to caller               
*                                                                      
* SECTION: Begin                                                        
*                                                                      
* PURPOSE: Tell the operator that we have been loaded into storage. 
*                                                                      
* OUTGOING REGS: 
*    R15=0   - OK                                                     
*    R15=4   - Error                                                  
*                                                                      
BEGIN  SUBRT TYPE=BEGIN,PHASE=NO                                      

WTO ‘CGI$DA has been loaded...’                                
SUBRT TYPE=END,RC=0             RETURN TO CALLER               

DC
*                                                                      
* SECTION: FINISH                                                       
*                                                                      
* PURPOSE: Tell the operator that this CGI is being removed from        
*          storage. 
*                                                                      
FINISH SUBRT TYPE=BEGIN,PHASE=NO                                      

WTO ‘CGI$DA has been unloaded...’                              
SUBRT TYPE=END,RC=0                                            

*                                                                      
* SECTION: OPEN                                                         
* PURPOSE: This CGI has just been invoked. We’ll setup some fields       
*          and put the passed data into a dynamic web page that         
*          “READ” will use later on. 
*                                                                      
* Incoming: R0=Length of incoming data                                  
*        R1=Pointer to the FRBLOK

R2=Data area pointer                                        
*                                                                      
* OUTGOING REGS                                                         
*    R1=  Address to pass to “READ/CLOSE”                               
*    R15=0   - OK                                                      
*    R15=4   - Error                                                    
*                                                                      
OPEN   SUBRT TYPE=BEGIN,PHASE=NO   Start of the routine           

L    R8,4(R13)            We’ll need this for later       
L    R0,=F’4096’          Get some length                
GETVIS ,                   Allocate an ACBLOK             
LTR  R15,R15             Did it work ? 
BZ   EXITFINE           Yes...proceed                  
WTO  ‘CGI$DA: GETVIS Failed!’                              
LA   R15,8               Indicate a problem             
B    EXITOPEN           And get out                   

*                                                                      
* For re-entrancy, let’s setup a parameter list for CALL$CGI            
*                                                                      
EXITFINE LA   R2,16(R1)         Point to the buffer            

ST   R2,4(R1)            And save it                   
LA   R2,12(R1)         Point to the pointer           
ST   R2,8(R1)          Save this value                

*                                                                      
EXITOPEN ST   R1,SVER01       Pass back the pointer          

SUBRT TYPE=END,RC=0(R15)  Return to caller               
*                                                                      
* SECTION: CLOSE                                                        
* PURPOSE: Free up previously allocated storage. 
*                                                                      
* Incoming:R2=ACBLOK address                                           
*           R13=Savearea                                                
*           R15=Base                                                   
*                                                                      
* OUTGOING REGS: 
*    R15=0   - OK, CLOSE WORKED                                         
*    R15=4   - CLOSE FAILED                                             
*                                                                      
CLOSE  SUBRT TYPE=BEGIN,PHASE=NO                                   

L     R0,=F’4096’         Get the length                 
LR    R1,R2               Get the pointer                
FREEVIS ,                  And release storage            
SUBRT RC=0(R15)          Return to caller               
END  

CONTINUED

The difference between CALL$DA and
CALL$DAH is minimal. I could have
changed CALL$DA to return HTML code
if a bit indicated that this was an HTML
request instead of a text request, but I am
presenting this as a simplified way of doing
a quick conversion without affecting any
other application that may be using the
called subroutine.

None of the other routines (e.g., DSPLHEX,
CNVTTSTCK or TIMEDIFF) need to be
changed since they don’t do anything with
the formatting of the data. Remember, a
CGI is nothing more than another program.

CREATING THE CGI

There are two methods of CGI processing
with TCP/IP for VSE. You can either have
the CGI run as a CICS application or you
can have it run in a batch partition. There
are advantages to each method. If you run it
within CICS, you have access to all of the
CICS data areas with all of the associated
security involved. If you run it in a batch
partition, you have the advantage of being
able to access data areas that are not defined
to CICS. There are also performance issues
related to each.

CGI AS AN ONLINE PROGRAM

This is probably the simplest of conversions
possible. If the program that you are converting
is already a CICS transaction, TCP/IP for
VSE provides a Command-Level pre-
processor to convert all CGI responses in
the following manner:

EXEC CGI SEND FROM(name) LENGTH(length) 

Where:

● Name: The name of an outbound buffer
● Length: The length of the buffer 

This can be done in any CICS-supported
language. The program can also transfer
control to other transactions that also issue
“EXEC CGI” requests, thus allowing multiple
transactions to contribute to the completed
web page. You can quickly and easily take
any CICS transaction that would normally
do a screen request to use this command to
send all of the data back to the web browser.
As with the example shown in Figure 2, you
will need to imbed HTML commands for
fancy formatting of the screen. This is one
of the features that the CAF (CICS Access
Facility) feature of TCP/IP for VSE offers.
You can also have that same CICS trans-
action issue either screen responses or CGI
responses, depending on how that CICS
transaction was invoked. In this way, a single
transaction can perform both local and
Internet functions. 

Since $DA was not a CICS transaction, I
will demonstrate how to convert non-CICS
programs into a CGI. This requires some
more coding and is limited to Assembler
and REXX.

CGI AS A BATCH PROGRAM

First, to define this application to TCP/IP
for VSE, you will need to include the fol-
lowing in your TCP/IP startup, or issue the
command manually from the console or
batch utility:

DEFINE FILE,PUBLIC=’$DA’,DRIVER=CGI$DA,
TYPE=CGI

This will cause CGI$DA to be invoked
every time a “$DA” request comes across
the web.

Next, you will need to assemble the code
shown in Figure 2. One of the great features
of this CGI is that it can be a generic
skeleton for any future CGI writing that you
do. All of the logic concerning the building
and formatting of data is done outside of
this code. This means that you can change
the “CALL CALL$DAH” to point to any
other routine. In fact, you can also have the
following commands:

DEFINE FILE,PUBLIC=’$DA’,DRIVER=CGI$DA,
TYPE=CGI

DEFINE FILE,PUBLIC=’STATUS’,DRIVER=

CGI$DA,TYPE=CGI

By looking at the data passed in the
FRBLOK, you could determine which
“public” name was used to call the CGI and
call a different subroutine based on what
was passed in field FRDPIE1. 

CONCLUSION

At this point, we’ve taken a simple appli-
cation that was meant for a 3270 session and
converted it to include HTML data within
the returned data so that it can easily be
viewed by the web. We then built a generic
CGI that passes control to that modified
routine and returns the result to the web
browser. While the basic CGI required some
additional coding, it can now be used for
migrating other applications in this manner.
The primary application, however, required
very little modification.

However, there are times when converting
a program to a CGI is not appropriate and it
is better that the converted program be a
server rather than a CGI. For example, an
HTTP session is not consistent (after a file
is transferred the session is terminated),
which presents problems with security and
general interactivity. Because of that, Part
III will examine how to convert $DA into a
multi-user server. 

This series is not about
using screen-scraper 

programs or providing
3270-access through the
web; rather it is about
eliminating the use of 

a 3270 client altogether.

At this point, we’ve taken 
a simple application that

was meant for a 3270 session
and converted it to include

HTML data within the
returned data so that it can
easily be viewed by the web.


